

FT64F0AX

MSCK Application note

目录

1. 慢时钟测量	3
1.1. 慢时钟测量相关寄存器汇总.....	3
1.2. 测量原理.....	4
1.3. 上电自动测量.....	4
1.4. 操作步骤.....	6
2. 应用范例.....	7
联系信息	13

FT64F0Ax MSCK 应用

1. 慢时钟测量

芯片集成了两个内部 RC 振荡器，一个是经过出厂校准的高速高精度的 16M 快时钟 HIRC，一个是低功耗的 32K 时钟 LIRC，利用慢时钟测量功能可以把 LIRC 的周期用系统时钟计算出来。此功能可以比较精准的测量内部慢时钟周期。

1.1. 慢时钟测量相关寄存器汇总

名称	状态	寄存器	地址	复位值
CKMAVG	<u>LIRC 和 HIRC 交叉校准时 4 次平均测量模式</u> 1 = 使能 0 = <u>关闭</u>	MSCKCON[1]	0x41D	RW-0
CKCNTI	<u>启动 LIRC 和 HIRC 的交叉校准功能</u> 1 = <u>启动</u> 0 = 完成 (测量完毕自动清零)	MSCKCON[0]		RW-1
SOSCPR	<u>校准 LIRC 周期所需的 HIRC 周期数</u>	SOSCPR[11:0]	0x41F[3:0] 0x41E[7:0]	RW-FFF

表 1-1 LIRC 和 HIRC 交叉校准控制/状态位

名称	状态		寄存器	地址	复位值
GIE	全局中断	1 = 使能 (PEIE, CKMIE 适用) 0 = <u>全局关闭</u> (唤醒不受影响)	INTCON[7]	Bank 首地址 +0x0B	RW-0
PEIE	外设总中断	1 = 使能 (CKMIE 适用) 0 = <u>关闭</u> (无唤醒)	INTCON[6]		RW-0
CKMIE	LIRC 和 HIRC 交叉校准完成中断	1 = 使能 0 = <u>关闭</u> (无唤醒)	PIE1[1]	0x91	RW-0
CKMIF ¹	LIRC 和 HIRC 交叉校准完成标志位	1 = Yes (锁存) 0 = <u>No</u>	PIR1[1]	0x11	R_W1C-0

表 1-2 LIRC 和 HIRC 交叉校准中断使能/状态位

¹ 写 1 清 0，写 0 无效。建议只使用 STR、MOVWI 指令进行写操作，而不要用 BSR 或 IOR 指令。

1.2. 测量原理




图 1-1 慢时钟测量模式原理框图

慢时钟测量类似于定时器的捕捉模式，处于这种模式下，被测量时钟 LIRC 的边沿（任意沿）将会触发定时器，在另一高速时钟（如 HIRC）的作用下开始计数，在此后的第 2 个（或第 8 个，平均模式时）LIRC 边沿到来时，定时器停止计数，同时把定时器的值锁存到 SOSCPRH/L 寄存器。

慢时钟测量使用的定时器是 TIM2。

注意：

1. 在慢时钟测量过程中软件不要写 SOSCPRH/L；
2. 不要在单步调试下做慢时钟测量，因为暂停模式下 TIM2 被停止，这样会导致测量结果不正确；
3. 若 SYSON = 0 时，慢时钟测量无法在 SLEEP 模式下进行，不要在测量运行时进入 SLEEP 模式。

1.3. 上电自动测量

在上电后慢时钟测量将会自动启动，此时 CKCNTI 置 1，CKMAVG 为 0，打开 LIRC 和 HIRC。TIM2 的时钟被自动配置为 16M 的内部高速时钟，即类似设置 T2CKSRC 为 001、TIM2EN=1 的功能，但未配置这些位。TIM2 使用默认配置，无需置位 CEN 使能 TIM2 计数，此时不能配置 TIM2。

在测量过程中，实际应用程序已在运行，若要使用 TIM2 则需要查询 CKCNTI。若 CKCNTI 为 0 即可使用 TIM2，此时 SOSCPR 寄存器的值为有效值，其单位为 F_{HSI} 时钟的个数。

注意：上电自动测量不会置位 CKM 中断标志。




图 1-2 上电慢时钟自动测量时序图

1.4. 操作步骤

1. 为提高计量精度，建议设置 T2CKSRC = 001，TIM2EN=1，选择 16M 的内部高速时钟；
2. 关闭 TIM2 的相关中断使能，设置 TIM2ARRH/L 为最大值，设置 TIM2PSC = 0000；
3. 设置 TIM2CR1 为复位值，再将 CEN 置 1，使能 TIM2；
4. 如果选择 4 次平均，则把 MSCKCON.1 置 1，否则把它清 0；
5. 置位 MSCKCON.0，开始测量；
6. 测量结束后 MSCKCON.0 自动清 0，中断标志置 1；
7. 可以用查询或中断的方式等待结束；
8. 当查询到中断标志为 1 时读取得到的 SOSCPR 即为最终结果。




图 1-3 慢时钟测量模式时序图

2. 应用范例

```
//=====
/* 文件名: ASM_64F0Ax_MSCK.ASM
* 功能:   FT64F0Ax_MSCK 功能演示
* IC:    FT64F0A5    TSSOP20
* 内部:   16M/2T
* 说明:   慢时钟测量完成后, 将在 PA4 输出高电平
*         在地址为 40,41 的 SRAM 上写入测量值(开启 4 次平均模式)
*         不开平均模式慢时钟频率=16M/TestTime
*         开平均模式慢时钟频率=16M/TestTime/4
*
*         FT64F0A5  TSSOP20
*         -----
* NC-----|1(PA5)      (PA4)20|--DemoPortOut
* NC-----|2(PA6)      (PA3)19|-----NC
* NC-----|3(PA7)      (PA2)18|-----NC
* NC-----|4(PC0)      (PA1)17|-----NC
* NC-----|5(PC1)      (PA0)16|-----NC
* NC-----|6(PB7)      (PB0)15|-----NC
* GND-----|7(GND)     (PB1)14|-----NC
* NC-----|8(PB6)      (PB2)13|-----NC
* VDD-----|9(VDD)     (PB3)12|-----NC
* NC-----|10(PB5)     (PB4)11|-----NC
*
*         -----
*/
//=====
#include <FT64F0AX.INC>;
;=====
;RAM DEFINE
;=====
TEMP_SOSCPRL EQU 0X40
TEMP_SOSCPRH EQU 0X41

W_TMP EQU 0X70
S_TMP EQU 0X71
;=====
;CONSTANT DEFINE
;=====
PIE1_DEF EQU B'00000010' ;允许慢时钟测量中断
INTCON_DEF EQU B'11000000' ;使能全局中断和外设中断
OSCCON_DEF EQU B'01110001' ;16MHz 1:1

WPUA_DEF EQU B'00000000' ;弱上拉的开关, 0-关, 1-开
WPUB_DEF EQU B'00000000'
```

```

WPUC_DEF      EQU      B'00000000'

WPDA_DEF      EQU      B'00000000'      ;弱下拉的开关, 0-关, 1-开
WPDB_DEF      EQU      B'00000000'
WPDC_DEF      EQU      B'00000000'

TRISA_DEF     EQU      B'00000000'      ;输入输出设置, 0-输出, 1-输入
TRISB_DEF     EQU      B'00000000'      ;PA4-OUT
TRISC_DEF     EQU      B'00000000'

PSRC0_DEF     EQU      B'11111111'      ;源电流设置最大
PSRC1_DEF     EQU      B'11111111'
PSRC2_DEF     EQU      B'00001111'

PSINK0_DEF    EQU      B'11111111'      ;灌电流设置最大
PSINK1_DEF    EQU      B'11111111'
PSINK2_DEF    EQU      B'00000011'

ANSELA_DEF    EQU      B'00000000'      ;设置对应的 IO 为数字 IO

PCKEN_DEF     EQU      B'00000100'      ;使能 TIM2 时钟模块
TCKSRC_DEF    EQU      B'00010000'      ;TIM2 时钟源 HIRC
;=====
;USER DEFINE
;=====
#define      DemoPortOut      PORTA,4
;=====
;PROGRAM START
;=====
      ORG      0x0000
      LJUMP    RESTART
      ORG      0x0004
      LJUMP    INT_PROGRAM
;=====
;INT_PROGRAM
;=====
INT_PROGRAM:
      STR      W_TMP
      SWAPR    STATUS,W
      STR      S_TMP
INT_RET:
      BANKSEL  PORTA
      BSR      PORTA,4      ;PA4 置高, 表示慢时钟测量已完成
      BANKSEL  SOSCPR

```



```

LDR        SOSCPRL,0
BANKSEL    TEMP_SOSCPRL
STR        TEMP_SOSCPRL    ;将慢时钟测量值低 8 位放在地址为 40 的 SRAM 上
BANKSEL    SOSCPRH
LDR        SOSCPRH,0
BANKSEL    TEMP_SOSCPRH
STR        TEMP_SOSCPRH    ;将慢时钟测量值高 4 位放在地址为 41 的 SRAM 上
BANKSEL    PIR1
BSR        PIR1,CKMIF      ;清除慢时钟测量完成中断标志位

SWAPR      S_TMP,0
STR        STATUS
SWAPR      W_TMP,1
SWAPR      W_TMP,0
RETI

;=====
;SYSTEM START
;=====
RESTART:
    BANKSEL    PORTA
    LCALL      INITIAL
    LCALL      MSCK_CONFIG

MAIN_LOOP:
    NOP
    NOP
    LJUMP      MAIN_LOOP

;=====
;INITIAL
;=====
INITIAL:
    BANKSEL    OSCCON
    LDWI       OSCCON_DEF
    STR        OSCCON

    BANKSEL    INTCON
    LDWI       INTCON_DEF
    STR        INTCON

    BANKSEL    PORTA
    LDWI       0X00
    STR        PORTA
    STR        PORTB
    STR        PORTC

```

BANKSEL	TRISA
LDWI	TRISA_DEF
STR	TRISA
LDWI	TRISB_DEF
STR	TRISB
LDWI	TRISC_DEF
STR	TRISC

BANKSEL	WPUA
LDWI	WPUA_DEF
STR	WPUA
LDWI	WPUB_DEF
STR	WPUB
LDWI	WPUC_DEF
STR	WPUC

BANKSEL	WPDA
LDWI	WPDA_DEF
STR	WPDA
LDWI	WPDB_DEF
STR	WPDB
LDWI	WPDC_DEF
STR	WPDC

BANKSEL	PSRC0
LDWI	PSRC0_DEF
STR	PSRC0
LDWI	PSRC1_DEF
STR	PSRC1
LDWI	PSRC2_DEF
STR	PSRC2

BANKSEL	PSINK0
LDWI	PSINK0_DEF
STR	PSINK0
LDWI	PSINK1_DEF
STR	PSINK1
LDWI	PSINK2_DEF
STR	PSINK2

BANKSEL	ANSELA
LDWI	ANSELA_DEF
STR	ANSELA

```

    BANKSEL    PIR1
    BSR        PIR1,CKMIF
    BANKSEL    PIE1
    LDWI       PIE1_DEF
    STR        PIE1

    BANKSEL    PCKEN
    LDWI       PCKEN_DEF
    STR        PCKEN
    BANKSEL    TCKSRC
    LDWI       TCKSRC_DEF
    STR        TCKSRC
,*****Clear SRAM*****
    BANKSEL    PORTA
    LDWI       0X00
    STR        FSR0H
CLEAR_RAM_BANK0:
    LDWI       20H
    STR        FSR0L
CLEAR_RAM_BANK0_LOOP:
    CLRR       INDF0
    INCR       FSR0L,F
    LDWI       80H
    XORWR      FSR0L,W
    BTSS       STATUS,Z
    LJUMP      CLEAR_RAM_BANK0_LOOP
CLEAR_RAM_BANK1:
    LDWI       0A0H
    STR        FSR0L
CLEAR_RAM_BANK1_LOOP:
    CLRR       INDF0
    INCR       FSR0L,F
    LDWI       00H
    XORWR      FSR0L,W
    BTSS       STATUS,Z
    LJUMP      CLEAR_RAM_BANK1_LOOP
    INCR       FSR0H,F
CLEAR_RAM_LOOP:
    LDWI       10
    SUBWR      FSR0H,W
    BTSS       STATUS,0
    LJUMP      CLEAR_RAM_BANK0
    RET

```

```
=====
;MEASURE
;=====
MSCK_CONFIG:
    BANKSEL    TIM2ARRH
    LDWI       0XFF
    STR        TIM2ARRH
    LDWI       0XFF
    STR        TIM2ARRL
    LDWI       0X00
    STR        TIM2PSCR
    BANKSEL    TIM2CR1
    BSR        TIM2CR1,T2CEN    ;使能 Timer2 计数器
    BANKSEL    MSCKCON
    BSR        MSCKCON,CKMAVG   ;打开平均模式
    BSR        MSCKCON,CKCNTI   ;使能快时钟测量慢时钟周期
    RET

END
```

联系信息**Fremont Micro Devices (SZ) Corporation**

#5-8, 10/F, Changhong Building
Ke-Ji Nan 12 Road, Nanshan District,
Shenzhen, Guangdong, PRC 518057

Tel: (+86 755) 8611 7811

Fax: (+86 755) 8611 7810

Fremont Micro Devices (HK) Corporation

#16, 16/F, Block B, Veristrong Industrial Centre,
34-36 Au Pui Wan Street, Fotan, Shatin, Hong Kong SAR

Tel: (+852) 2781 1186

Fax: (+852) 2781 1144

<http://www.fremontmicro.com/>

* Information furnished is believed to be accurate and reliable. However, Fremont Micro Devices (SZ) Corporation assumes no responsibility for the consequences of use of such information or for any infringement of patents of other rights of third parties, which may result from its use. No license is granted by implication or otherwise under any patent rights of Fremont Micro Devices (SZ) Corporation. Specifications mentioned in this publication are subject to change without notice. This publication supersedes and replaces all information previously supplied. Fremont Micro Devices (SZ) Corporation products are not authorized for use as critical components in life support devices or systems without express written approval of Fremont Micro Devices (SZ) Corporation. The FMD logo is a registered trademark of Fremont Micro Devices (SZ) Corporation. All other names are the property of their respective owners.